# $Id: PKGBUILD 67396 2012-03-11 10:52:58Z lcarlier $
# Maintainer: Kaiting Chen <kaitocracy@gmail.com>
# Contributor: Gergely Imreh <imrehgATgmailDOTcom>

pkgname=libx86emu
pkgver=1.1
_pkgver=$pkgver-15.3
pkgrel=29
pkgdesc='x86 emulation library'
arch=('i686' 'x86_64')
url='http://www.opensuse.org/'
license=('BSD')
depends=('glibc')
makedepends=('rpmextract')
source=("http://download.opensuse.org/source/factory/repo/oss/suse/src/$pkgname-$_pkgver.src.rpm")
md5sums=('ba6a0fc46307dd79078579140aab2f02')

build() {
  cd "${srcdir}"

  rpmextract.sh "${pkgname}-${_pkgver}.src.rpm"
  tar -xjf $pkgname-$pkgver.tar.bz2

  cd "${srcdir}/${pkgname}-${pkgver}"

  make LIBDIR=/usr/lib
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  make LIBDIR=/usr/lib DESTDIR=$pkgdir install

  install -Dm644 LICENSE \
  ${pkgdir}/usr/share/licenses/$pkgname/LICENSE
}
